

# Un oscillateur délivrant des signaux carré, triangulaire et sinusoïdal

J'invite le lecteur à consulter le site pour des informations complémentaires.  
 Page d'accueil du site Internet : [page d'accueil](#) d'autres pdf, sur différents sujets : [liste des PDE](#)



Ce document montre une façon de réaliser un oscillateur. La structure proposée permet de générer 3 formes d'ondes synchrones, comme le permet des GBF classiques de laboratoire.

## 1) L'oscillateur à relaxation de base

### Description :

- un intégrateur inverseur,
- et un comparateur « tout ou rien ».

Posons :  **$u_i(t)$  la sortie de l'intégrateur,**  
 **$V_s$  la sortie du comparateur.**



Rappelons que le comparateur tout ou rien est un montage dont la sortie  $V_s$  ne peut prendre que 2 valeurs possibles, ici,  $-V_{sat}$ , et  $+V_{sat}$ , tension bornée par l'alimentation bipolaire (non représentée ici) du circuit.  
 Sa sortie change d'état quand  $\varepsilon = e^+ - e^-$  passe à 0. Dans notre schéma,  $\varepsilon = e^+$ .


L'expression du potentiel  $e^+$  est :  $u_i(t) \frac{R_2}{R_1 + R_2} + V_s \frac{R_1}{R_1 + R_2}$ , avec  $V_s = +V_{sat}$  ou  $-V_{sat}$ .

Si  $V_s$  est à  $+V_{sat}$ , la commutation aura lieu quand :

$$u_i(t) \frac{R_2}{R_1 + R_2} + V_{sat} \frac{R_1}{R_1 + R_2} = 0, \text{ soit } u_i(t) = -V_{sat} \frac{R_1}{R_2}$$

Si  $V_s$  est à  $-V_{sat}$ , la commutation aura lieu quand :  $u_i(t) = +V_{sat} \frac{R_1}{R_2}$

Ce qui se traduit par le cyclogramme ci-contre :



### Fonctionnement


Prenons, comme condition initiale, la transition de  $-V_{sat}$  à  $+V_{sat}$  sur la sortie  $V_s(t)$ .

À  $V_s = V_{sat}$ , l'intégrateur inverseur délivre :

$$u_i(t) = -\frac{1}{RC} \int_0^t V_s(t) dt + u_i(0) = -\frac{V_{sat}}{RC} t + u_i(0).$$

Avec  $u_i(0) =$  valeur de  $u_i$  à  $t = 0+ = V_{sat} R_1 / R_2$

$u_i(t)$  est une **rampe décroissante**.



$$\text{On a : } e^+(t) = \left[ -\frac{V_{sat}}{RC} t + \frac{V_{sat} R_1}{R_2} \right] \frac{R_2}{R_1 + R_2} + V_{sat} \frac{R_1}{R_1 + R_2} = -\frac{R_2}{R_1 + R_2} \frac{V_{sat}}{RC} t + V_{sat} \frac{2 R_1}{R_1 + R_2} :$$

Quand  $e^+$  atteint 0, le comparateur bascule vers  $-V_{sat}$ , ce qui impose :

- une valeur négative à  $e^+$ , par le pont diviseur  $R_1, R_2$ ,
- un changement de sens de la sortie de l'intégrateur, qui délivre une **rampe croissante**

$$\text{Durant cette phase } e^+(t) = \frac{R_2}{R_1 + R_2} \frac{V_{sat}}{RC} t - V_{sat} \frac{2 R_1}{R_1 + R_2} .$$

## Loi donnant la période en fonction des éléments

Avec l'origine des temps posée lors de la commutation  $-V_{\text{sat}} + V_{\text{sat}}$ , la commutation suivante a lieu quand  $e^+$  passe par 0, soit à  $T/2$ .

$$\text{Il en résulte : } e^{+(T/2)} = -\frac{R_2}{R_1 + R_2} \frac{V_{\text{sat}}}{R C} \frac{T}{2} + V_{\text{sat}} \frac{2 R_1}{R_1 + R_2} = 0$$

$$\text{On déduit directement : } T = 4 R C \frac{R_1}{R_2}, \quad \text{soit } f = \frac{R_2}{4 R C R_1}$$

## 2) Simulations de l'oscillateur délivrant carre, triangle

On choisit d'utiliser les A. Op. de type LF411, pour l'intégrateur et le comparateur.

Les valeurs numériques des composants sont :

$$R_1 = 6,8 \text{ k}\Omega ; R_2 = 10 \text{ k}\Omega ;$$

$$R = 33 \text{ k}\Omega ; C = 10 \text{ nF}.$$

On attend une valeur de période de :

$$T = 4 \cdot 33\text{k} \cdot 10\text{n} \cdot \frac{6,8\text{k}}{10\text{k}} = 897,6 \text{ }\mu\text{s}.$$

Ainsi, un run de quelques ms permet d'avoir quelques périodes.

On impose une condition initiale pour être dans un cycle dès  $t = 0$ . Le début de simulation ne montre pas le régime établi, mais le lancement de l'oscillation suite aux conditions initiales.

### Oscillateur a relaxation

\* fichier oscillateur.cir

.lib eval.lib

\* circuit :

Vplus 15 0 DC 15 ; pour les LF411

Vmoins 14 0 DC -15

R VS 1 33k

C 1 ui 10n

X1 0 1 15 14 ui LF411 ; integrateur

R1 ui 2 6.8k

R2 2 VS 10k

X2 2 0 15 14 VS LF411 ; comparateur

.IC V(VS)=-15

.TRAN 1u 4m 2m 1u

.PROBE

.END


*Netlist, prête à simuler*

## Résultats

On mesure une période de  $901 \text{ }\mu\text{s}$ , soit une fréquence de  $1,11 \text{ kHz}$ .

$u_i$  évolue entre  $\pm 9,66 \text{ V}$ .

$V_s$  évolue entre  $\pm 14,177 \text{ V} = \pm V_{\text{sat}}$



Dans la partie croissante, l'équation théorique de  $u_i$  est :  $\frac{V_{\text{sat}}}{R C} t + u_i(0)$ .

Le coefficient directeur a donc pour valeur attendue :  $\frac{14,177}{33 \text{ k} 10 \text{ n}} = 0,0429 \text{ V} / \mu\text{s}$ .

## Interprétation :

On retrouve la pente en sortie de l'intégrateur :  $19,32 \text{ V} / 450,5 \text{ }\mu\text{s}$ , soit  $0,0429 \text{ V} / \mu\text{s}$ .

La période issue de la simulation est de quelques  $\mu\text{s}$  plus lente que celle calculée. Cet écart est dû au temps de montée et de descente du comparateur non nul, dû au slew rate du LF411.

### 3) Le conformateur à diodes


Pour délivrer un signal sinusoïdal à partir de cet oscillateur, on peut utiliser un passe bande très sélectif qui isole la fréquence fondamentale et élimine les autres harmoniques. Ce filtre peut être placé en  $u_i(t)$ . Cette solution présente l'inconvénient d'être figée à une fréquence particulière et précise.

Une autre solution est d'exploiter un conformateur à diodes :


#### Principe

Ce montage permet, à partir d'un signal triangulaire symétrique, de générer un signal défini par morceau, grâce à la mise en conduction progressive de diodes. L'objectif est de décrire une sinusoïde par tronçons.

Raisonnons sur 2 diodes : par symétrie, on peut travailler sur la partie supérieure ( $D_1, D_2$  etc), avec la tension  $V(in)$  positive, puis sur la partie inférieure ( $D'_1, D'_2$  etc), avec la tension  $V(in)$  négative.



Une diode peut être idéalisée par la caractéristique ci-contre, qui présente une tension de déchet  $V_d$  que l'on va supposer constante, et d'une résistance directe nulle.




On peut faire « glisser »  $V_d$  de chaque diode vers les sources de tensions  $E_1, E_2...$  en choisissant leur valeur numérique décalée d'autant et raisonner alors avec des diodes agissant en interrupteur parfait.

Dans ce cas, il vient simplement le fonctionnement suivant à  $V(in) > 0$  :


- à  $V(in) < E_1$ , aucune diode n'est passante. On a  $V(out) = V(in)$ .

- pour une certaine valeur de  $V(in)$ , la diode  $D_1$  est passante. Cela se produit pour  $V(out) > E_1$ . On a le schéma équivalent suivant : On a alors, par contribution de chaque source :

$$V(out) = V(in) \frac{R_1}{R + R_1} + E_1 \frac{R}{R + R_1}$$




- pour une certaine valeur de  $V(in)$ , la diode  $D_2$  est également passante. Cela se produit pour  $V(out) > E_2$ . On pose  $R_{12} = R_1/R_2$        $R_{10} = R_1/R$        $R_{20} = R_2/R$



On a alors, par contribution de chaque source :  $V(out) = V(in) \frac{R_{12}}{R + R_{12}} + E_1 \frac{R_{20}}{R_{20} + R_1} + E_2 \frac{R_{10}}{R_{10} + R_2}$


On construit ainsi une forme d'onde de la forme suivante :

- Zone (1) :  $V(out) = V(in)$ .
- Zone (2) :  $D_1$  ON :  $V(out) = \alpha V(in) + V_1$
- Zone (3) :  $D_1 D_2$  ON :  $V(out) = \beta V(in) + V_2$



Il faut donc des valeurs numériques judicieuses aux tensions et résistances. Nous allons voir qu'un nombre de 4 branches (2 fois 4 diodes) permet d'obtenir un résultat satisfaisant.



On a :  $E_1 < E_2 < E_3 < E_4$ .



C'est ce que propose le **circuit intégré ICL8038**, qui délivre carré, triangulaire et sinusoïde, dont le schéma du conformateur « à diodes » est donné ci-contre.

**Explications**

Les potentiels E<sub>1</sub>, E<sub>2</sub>, E<sub>3</sub>, etc sont réalisés par un pont diviseur R<sub>32</sub>, R<sub>33</sub>, R<sub>34</sub> etc. Les résistances R, R<sub>1</sub>, R<sub>2</sub> sont ici R<sub>44</sub>, R<sub>21</sub>, R<sub>20</sub> etc. Chaque diode est remplacée par l'association de 2 transistors. Le schéma complet est donc équivalent à 2 fois 4 diodes. Le fonctionnement est similaire, comme montré ici, pour une diode :




La conduction de la diode correspond à l'état passant de Q<sub>41</sub>. Le potentiel E<sub>1</sub> correspond au point commun R<sub>35</sub>, R<sub>36</sub>, qui est recopié à l'émetteur de Q<sub>41</sub>.

**Test du conformateur à diodes**

Dans notre application (triangle ± 9,66 V), les valeurs numériques des résistances ne peuvent pas être celles du circuit ICL8038. Elles sont donc ici différentes.

Pour tester ce conformateur, on réalise un montage qui exploite un signal triangulaire issu d'une source parfaite. On choisit une fréquence de 1 kHz (très proche de celle de notre oscillateur). Pour quantifier le taux de distorsion de façon précise, on utilise la commande « .four ».

Le conformateur est déclaré dans un **sous-circuit**.



La sortie du conformateur porte le numéro de nœud 3

conformateur à diodes  
\* fichier conformateur.cir  
\* circuit :

```
.lib eval.lib
Vplus 15 0 DC 15
Vmoins 14 0 DC -15
Vtriangle ui 0 pulse (-9.66 9.66 0 0.5m 0.5m 0 1m)
X3 ui 3 15 14 conformateur
```


```
.TRAN 0.1u 2m 0 0.1u
.four 1k V(3)
.SUBCKT conformateur entree sortie vplus vmoins
R entree sortie 68k
RA sortie 20 47k
RB sortie 21 100k
RC sortie 22 390k
D1 sortie 23 D1N4148
D2 20 24 D1N4148
D3 21 25 D1N4148
D4 22 26 D1N4148
R5 vplus 23 910
R4 23 24 47
R3 24 25 100
R2 25 26 150
R1 26 0 330
Rp1 0 27 330
Rp2 27 28 150
Rp3 28 29 100
Rp4 29 30 47
Rp5 30 vmoins 910
D5 27 22 D1N4148
D6 28 21 D1N4148
D7 29 20 D1N4148
D8 30 sortie D1N4148
.ENDS
.PROBE
.END
```

*Netlist, prête à simuler*

### Résultats

En visualisant la sortie, on « ne voit pas » de défaut apparent : allure sinusoïdale, pas de point anguleux.


Le résultat de la directive .four, donné dans le fichier de sortie, donne un taux de distorsion, calculé par Pspice, de 0,88%.



TOTAL HARMONIC DISTORTION = 8.8427E-01 PERCENT

### 4) Simulations de l'oscillateur avec conformateur à diodes


On insère le sous-circuit dans le schéma précédent :



```


Oscillateur a relaxation
* fichier oscillateur_3_sorties.cir
.lib eval.lib
* circuit :
Vplus 15 0 DC 15 ; pour les LF411
Vmoins 14 0 DC -15
R VS 1 33k
C 1 ui 10n
X1 0 1 15 14 ui LF411 ; integrateur
R1 ui 2 6.8k
R2 2 VS 10k
X2 2 0 15 14 VS LF411 ; comparateur
X3 ui out 15 14 conformateur
.IC V(VS)=-15
.TRAN 1u 99.32m 2m 1u
*
.SUBCKT conformateur entree sortie vplus vmoins
R entree sortie 68k
RA sortie 20 47k
RB sortie 21 100k
RC sortie 22 390k
D1 sortie 23 D1N4148
D2 20 24 D1N4148
D3 21 25 D1N4148
D4 22 26 D1N4148
R5 vplus 23 910
R4 23 24 47
R3 24 25 100
R2 25 26 150
R1 26 0 330
Rp1 0 27 330
Rp2 27 28 150
Rp3 28 29 100
Rp4 29 30 47
Rp5 30 vmoins 910
D5 27 22 D1N4148
D6 28 21 D1N4148
D7 29 20 D1N4148
D8 30 sortie D1N4148
.ENDS
.PROBE
.END
    
```

On obtient sans surprise le bon fonctionnement :



L'amplitude de la sinusoïde est de 6,53 V,  
la fréquence est 1,11 kHz


Une autre façon de juger de la « qualité » de la forme d'onde est de comparer avec une sinusoïde parfaite, générée ici par Probe, en demandant le tracé de :  
 $6.53 * \sin(2 * \pi * 1109.7 * \text{time} - \pi / 4)$ .  
Les 2 traces sont en coïncidence.



*Netlist, prête à simuler*

## Mesure du taux de distorsion

Les paramètres de la commande .TRAN sont réglés pour avoir une résolution fréquentielle poussée lors de l'affichage du spectre par la commande FFT. On n'affiche ici que les 2 raies principales, sur les 3 sorties (carré, triangle, sinus). Pour la sortie sinus, les ordonnées sont sur une échelle log.



En zoomant, on peut mesurer les principales harmoniques suivantes :

	Fondamental 1,11 kHz	Harmonique 3,33 kHz	Harmonique 5,55 kHz
<b>VS : carré</b>	18,05 V	6,01 V	3,5857 V
<b>Ui : triangle</b>	7,8439 V	0,871 V	0,3116 V
<b>Out : sinus</b>	6,5279 V	33,1 mV	43,127 mV

Les signaux périodiques se décomposent en série de Fourier :

**carré** : Son fondamental a pour expression  $\frac{4}{\pi} V_{\text{sat}} = \frac{4}{\pi} 14,77 = 18,805 \text{ V}$ , la raie suivante (fréquence triple) au tiers, soit 6,268 V. Le carré n'étant pas parfait (slew rate), la simulation donne un petit écart.

**triangle** : Son fondamental a pour expression  $\frac{8}{\pi^2} U_{i_{\text{max}}} = \frac{8}{\pi^2} 9,66 = 7,83 \text{ V}$ , la raie suivante (fréquence triple) au neuvième, soit 0,87 V. Ces valeurs théoriques sont confirmées par la simulation.

**sinus** : le convertisseur à diodes délivre un fondamental de 6,53 V d'amplitude. Une lecture fine montre les raies suivantes, ce qui permet de chiffrer le taux de distorsion harmonique :

$$d = \frac{10^{-3} \sqrt{33,1^2 + 43,127^2 + 15,065^2 + \dots}}{6,5279^2} \approx 0,13 \%$$

Cette valeur est différente de celle obtenue par une source triangle parfaite : en fait, la sortie triangle de notre oscillateur présente des « pointes un peu arrondies » (en zoomant finement), ce qui réduit l'amplitude des harmoniques, et « facilite » ainsi la mise en forme sinusoïdale. La distorsion obtenue est donc très faible. Pour information, le THD est spécifié à 1 % typique pour l'ICL8038AC. On remarque qu'il n'y a pas d'harmonique à la fréquence 2,22 kHz.

## Influence de la température

La température modifie le comportement des semi-conducteurs. La partie la plus critique est le convertisseur à diodes, car si la température augmente, le seuil de conduction des diodes diminue, et donc la forme d'onde générée va s'éloigner de la sinusoïde. Un essai à  $T = 125 \text{ °C}$  a montré des harmoniques aux multiples de la fréquence d'oscillation, ce qui a donné un taux de distorsion de 0,2 %. Dans cette simulation, on n'a pas tenu compte de la dérive des valeurs des résistances, condensateur, ce qui aurait modifié la fréquence, et également la mise en forme, surtout par l'influence de  $R$ ,  $R_A$ ,  $R_B$ ,  $R_C$ . Bien sûr, d'autres phénomènes interviennent comme l'offset, le courant BIAS des LF411, mais cela n'a pas affecté la fréquence.

## Conclusion

Cette étude a montré le fonctionnement d'un oscillateur à relaxation. L'intérêt de disposer d'une sortie triangulaire est la mise en œuvre facile d'un convertisseur à diodes pour fabriquer une onde sinusoïdale. Les simulations ont permis de vérifier le bon fonctionnement de ce montage à 3 sorties.

articles 1 à 43 : sur le livre

**Tableau récapitulatif des articles PDF disponibles sur ce site**

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	<a href="#">présentation</a>	<a href="#">document PDF</a>
44	Exemples basiques et des exercices...	<a href="#">présentation</a>	<a href="#">document PDF</a>
45	Un exemple de circuit passif	<a href="#">présentation</a>	<a href="#">document PDF</a>
46	Un oscillateur Colpitts	<a href="#">présentation</a>	<a href="#">document PDF</a>
47	Compensation en fréquence des amplificateurs opérationnels	<a href="#">présentation</a>	<a href="#">document PDF</a>
48	Un amplificateur à transistors bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
49	Une bascule D Flip Flop CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
50	Une porte XOR à transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
51	Un VCO à 12 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
52	Une PLL à moins de 20 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
53	Un oscillateur à résistance négative	<a href="#">présentation</a>	<a href="#">document PDF</a>
54	Une charge électronique	<a href="#">présentation</a>	<a href="#">document PDF</a>
55	Un amplificateur en classe C	<a href="#">présentation</a>	<a href="#">document PDF</a>
56	Le monostable 74 123	<a href="#">présentation</a>	<a href="#">document PDF</a>
57	Un amplificateur en classe D	<a href="#">présentation</a>	<a href="#">document PDF</a>
58	Le transformateur en linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
59	La loi d'ohm thermique	<a href="#">présentation</a>	<a href="#">document PDF</a>
60	Le transformateur en non linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
61	Robustesse d'un oscillateur en anneau	<a href="#">présentation</a>	<a href="#">document PDF</a>
62	Une alimentation stabilisée	<a href="#">présentation</a>	<a href="#">document PDF</a>
63	Modélisation d'un haut-parleur	<a href="#">présentation</a>	<a href="#">document PDF</a>
64	Un synthétiseur de fréquence	<a href="#">présentation</a>	<a href="#">document PDF</a>
65	Un ampli audio de Sparkfun	<a href="#">présentation</a>	<a href="#">document PDF</a>
66	Simulation logique et analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
67	Un oscillateur à relaxation	<a href="#">présentation</a>	<a href="#">document PDF</a>
68	Lecteur de TAG RFID 125 kHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
69	Diagramme de l'œil avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
70	Un amplificateur hautes fréquences	<a href="#">présentation</a>	<a href="#">document PDF</a>
71	Une bizarrerie enfin expliquée...	<a href="#">présentation</a>	<a href="#">document PDF</a>
72	Comprendre le paramétrage de la FFT	<a href="#">présentation</a>	<a href="#">document PDF</a>
73	La relation de Bennett	<a href="#">présentation</a>	<a href="#">document PDF</a>
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	<a href="#">présentation</a>	<a href="#">document PDF</a>
75	Une horloge biphase sans recouvrement	<a href="#">présentation</a>	<a href="#">document PDF</a>
76	Quelques simulations sur la diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
77	Un ampli classe A, avec transformateur de sortie	<a href="#">présentation</a>	<a href="#">document PDF</a>
78	Des stimuli pour PSpice	<a href="#">présentation</a>	<a href="#">document PDF</a>
79	Simuler le TL431 : zener ajustable	<a href="#">présentation</a>	<a href="#">document PDF</a>
80	Un ADC flash	<a href="#">présentation</a>	<a href="#">document PDF</a>
81	Une chaîne d'acquisition : S&H, ADC, DAC	<a href="#">présentation</a>	<a href="#">document PDF</a>
82	Un amplificateur 50 MHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
83	Un dérivateur non inverseur	<a href="#">présentation</a>	<a href="#">document PDF</a>
84	Un amplificateur bipolaire avec push pull CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
85	Rôle des répéteurs logiques dans un circuit intégré	<a href="#">présentation</a>	<a href="#">document PDF</a>
86	Un driver logique CMOS pour charge 50 ohms	<a href="#">présentation</a>	<a href="#">document PDF</a>
87	Des triggers de Schmitt et des applications	<a href="#">présentation</a>	<a href="#">document PDF</a>
88	Un filtre gaussien analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
89	Un générateur de bruit rose	<a href="#">présentation</a>	<a href="#">document PDF</a>

90	Un anémomètre à fil chaud : simulation comportementale	<a href="#">présentation</a>	<a href="#">document PDF</a>
91	Un oscillateur à pont de Wien stabilisé par CTN	<a href="#">présentation</a>	<a href="#">document PDF</a>
92	L'emballement thermique d'une diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
93	Les puissances dans un amplificateur	<a href="#">présentation</a>	<a href="#">document PDF</a>
94	Asservissement de puissance dans une résistance	<a href="#">présentation</a>	<a href="#">document PDF</a>
95	Asservissement de la puissance émise par une antenne radio	<a href="#">présentation</a>	<a href="#">document PDF</a>
96	Un driver de LED de puissance	<a href="#">présentation</a>	<a href="#">document PDF</a>
97	Exploiter Pspice pour simuler des filtres numériques	<a href="#">présentation</a>	<a href="#">document PDF</a>
98	Un filtre en cosinus surélevé avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
99	Effet de la température sur un amplificateur en classe A	<a href="#">présentation</a>	<a href="#">document PDF</a>
100	Un amplificateur à transistors JFET et bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
<b>Supplément, hors article :</b>			
<b>mon cours « Electronique pour les communications numériques », <a href="#">polycopié couleur 201 pages en pdf</a></b>			

[retour à l'écran d'accueil de ce site](#)